

# Building test.check Generators

Gary Fredericks

# Who Am I?

- I am Gary Fredericks
- I live in Chicago
- I work (with Clojure) at DRW
- I sometimes work on improving `test.check`

- `test.check`

- `test.check`
  - Created 4 years ago by Reid Draper
  - Clojure library for property-based-testing
    - Haskell's QuickCheck
  - Inputs described by writing generators

- `test.check`
  - Created 4 years ago by Reid Draper
  - Clojure library for property-based-testing
    - Haskell's QuickCheck
  - Inputs described by writing generators
- `clojure.spec`

- `test.check`
  - Created 4 years ago by Reid Draper
  - Clojure library for property-based-testing
    - Haskell's QuickCheck
  - Inputs described by writing generators
- `clojure.spec`
  - Announced 18 months ago
  - Creates generators for basic specs
  - User-supplied generators for complex specs

- Generators, in General
- Building
- Fine-Tuning

# Generators, in General



# clojure.spec

c.spec.alpha

Making specs

c.spec.test.alpha

instrument, check, etc.

c.spec.gen.alpha

Dynaload!

# test.check

c.t.check

c.t.check.properties

Assembling/running tests

c.t.check.clojure-test

clojure.test integration

c.t.checks.generators

Building generators



# Generators, in General

## What's in the box?

# Basic Data Generators

```
1 (def generate-some-great-data
2   (gen/hash-map
3     :a-boolean          gen/boolean
4     :some-small-integers (gen/vector gen/nat)
5     :a-large-integer    gen/large-integer
6     :a-double           gen/double
7     :a-color            (gen/elements [:red :green :blue])
8     :a-uuid             gen/uuid
9     :a-string-and-a-keyword (gen/tuple gen/string
10                                     gen/keyword)))
11
12 (gen/generate generate-some-great-data 10)
13 =>
14 {:a-boolean          false,
15  :some-small-integers [2 5 5 5 10 5 0],
16  :a-large-integer    -6,
17  :a-double           0.47607421875,
18  :a-color            :green
19  :a-uuid             #uuid "a06e2893-6fcc-4b42-8e2f-ba5da58202ac",
20  :a-string-and-a-keyword ["ö" :eA:5C*:02]}
```

- `(gen/tuple g1 g2 ...)`
- `(gen/fmap (fn [x] x') g)`
- `(gen/bind g (fn [x] g'))`
- `(gen/such-that pred g)`
- `(gen/frequency [[w1 g1] [w2 g2] ...])`
- `(gen/one-of [g1 g2 ...])`

```
1 (defn gen-fav-number-assertion
2   []
3   (let [x (rand-int 10)]
4     (str "My favorite number is " x)))
5
6 (gen-fav-number-assertion)
7 => "My favorite number is 9"
```

```
1 (defn gen-fav-number-assertion
2   [size]
3   (let [x (rand-int size)]
4     (str "My favorite number is " x)))
5
6 (gen-fav-number-assertion 100)
7 => "My favorite number is 91"
```

## Size -> (Data, Shrinks Data)

```
1 (defn gen-fav-number-assertion
2   [size]
3   (let [x (rand-int size)]
4     [(str "My favorite number is " x)
5      ;; some sort of recursively lazy
6      ;; expression that generates a
7      ;; lazy tree of smaller strings
8     ]))
9
10 (gen-fav-number-assertion 100)
11 => ["My favorite number is 91" (...)]
```

## Rand Double -> Size -> (Data, Shrinks Data)

```
1 (defn gen-fav-number-assertion
2   [rng size]
3   (let [x (-> rng (rand/rand-double) (* size) (long)))]
4     [(str "My favorite number is " x)
5      ;; some sort of recursively lazy
6      ;; expression that generates a
7      ;; lazy tree of smaller strings
8     ]))
9
10 (gen-fav-number-assertion (rand/make-random 42) 100)
11 => ["My favorite number is 91" (...)]
```



```
1 (defn collection-of
2   "Returns a generator of a collection
3   with elements generated from the
4   supplied generator."
5   [gen]
6   (fn [rng size]
7     ;; and now for the tricky bit
8     [
9     ;; um
10    ]))
```

`clojure.core/rand-int`

vs

`gen/large-integer`

- Abstract sizing/growth
- Shrinking
- Functional Determinism

```
1 (gen/sample g <num-samples=10>)
2 => (data0 data1 data2 ...)
3
4 (gen/generate g <size=30>)
5 => data
```

- `gen/tuple`
- `gen/one-of`
- `gen/frequency`
- `gen/such-that`
- `gen/fmap`
- `gen/bind`



## Similarities with FP:

- Universality
  - The built-in generators can generate just about anything
  - But understanding how takes practice
- Circumventability
  - e.g.,
    - `rand`, etc.
    - `gen/sample`, `gen/generate`
    - using today's date, DB records
  - Undermines the value proposition
    - Abstract sizing/growth
    - Shrinking
    - Functional Determinism



# Combinators, this time in detail

- `(gen/tuple g1 g2 ...)`
- `(gen/fmap (fn [x] x') g)`
- `(gen/bind g (fn [x] g'))`
- `(gen/such-that pred g)`
- `(gen/frequency [[w1 g1] [w2 g2] ...])`
- `(gen/one-of [g1 g2 ...])`



# gen/tuple



gen/one-of





gen/frequency








gen/such-that




$(\triangle \Rightarrow T/F)$



gen/fmap





gen/bind



gen/fmap



gen/bind



Example:

Generate a non-empty collection  
and a random element from  
that collection.

## Example: collection-and-element

```
1 (def gen-collection-and-element
2   (gen/fmap (fn [coll]
3             ;; need to pick a random element
4             ;; somehow...
5             )
6   (gen/not-empty
7     (gen/vector gen/large-integer))))
```

## Example: collection-and-element

```
1 (def gen-collection-and-element
2   (gen/fmap (fn [coll]
3             ;; cheat!
4             [coll (rand-nth coll)])
5   (gen/not-empty
6     (gen/vector gen/large-integer))))
7
8 (gen/generate gen-collection-and-element 20)
9 [[-582 -159 -1 2 -29 -9 -18605 -2 961 -31 90] -9]
```

## Example: collection-and-element

```
1 (def gen-collection-and-element
2   (gen/fmap (fn [coll]
3             ;; pick a random element, then use
4             ;; gen/fmap to wrap it up with the
5             ;; collection
6             (gen/fmap (fn [x] [coll x])
7                       (gen/elements coll))))
8   (gen/not-empty
9     (gen/vector gen/large-integer))))
10
11 (gen/generate gen-collection-and-element)
12 #clojure.test.check.generators.Generator{:gen #function[clojure.test.check.generators/gen-collection-and-element]}
13 ;; we just generated a generator, not data
```



## Example: collection-and-element

```
1 (def gen-collection-and-element
2   (gen/bind (gen/not-empty
3             (gen/vector gen/large-integer))
4   (fn [coll]
5     ;; pick a random element, then use
6     ;; gen/fmap to wrap it up with the
7     ;; collection
8     (gen/fmap (fn [x] [coll x])
9               (gen/elements coll))))))
10
11 (gen/generate gen-collection-and-element 20)
12 [[-1 40 3 6199 -77 -433763 -412 46055 0 -6 0] 40]
```

## Example: collection-and-element

```
1 (def gen-collection-and-element
2   (gen/let [coll (gen/not-empty
3               (gen/vector gen/large-integer))
4             x     (gen/elements coll)]
5     [coll x]))
6
7 (gen/generate gen-collection-and-element 20)
8 [[-37830 3546 -210996 3 0 -35206 6 517097] 3546]
```

## Example: a 2d matrix

```
1  ;; Bad, because the inner vectors
2  ;; can have different lengths.
3  (gen/vector (gen/vector gen/large-integer))
4
5
6  ;; Generate a width first, then use bind to generate
7  ;; a collection of vectors with the same width.
8  (gen/let [width gen/nat
9            rows (gen/vector
10                 (gen/vector gen/large-integer
11                             width))]
12    rows)
```

# Summary: Generators, in General

- There are concrete data generators, and abstract combinators
- The abstractness is in service of the value proposition
- Like FP - familiarity takes practice

# Building

- Generate a random (nested) directory of files
- Generate changes to a directory

```
1 (def gen-file-name
2   (gen/such-that #(not (re-find #"/" %))
3                 gen/string-ascii))
4
5 (gen/sample gen-file-name)
6 (" " " " "F" "kI$" "O%" "T4\\:" "W'\\:P" "qS4" " " " ")
```

```
1 (gen/sample gen-file-name 1000)
2 ;; ExceptionInfo Couldn't satisfy such-that
3 ;; predicate after 10 tries.
4 ;; clojure.core/ex-info (core.clj:4725)
```

Common culprit in spec: (s/and s1 s2 s3 ...)



```
1 (def gen-file-name-2
2   (gen/fmap #(clojure.string/replace % "/" "")
3             gen/string-ascii))
4
5 (gen/sample gen-file-name-2)
6 (" " "0" "E" "=e+" "ozdH"
7  "_ " "WNL_6" "{Zqm" "CY(1I)H." "z")
```

```
1 (def gen-file-contents
2   gen/bytes)
3
4 (gen/sample gen-file-contents)
5 (#bytes ""
6  #bytes "0e"
7  #bytes "8b"
8  #bytes "80e1d5"
9  #bytes "8ad6"
10 #bytes ""
11 #bytes "b0a0224119c8"
12 #bytes "f39a97ff"
13 #bytes "8a"
14 #bytes "22908190fc09eca901")
```

```
1 (def gen-permissions-octal
2   (gen/fmap #(format "%03o" %)
3             (gen/large-integer {:min 0 :max 0777})))
4
5 (gen/sample gen-permissions-octal)
6 ("350" "330" "350" "017" "150"
7  "155" "666" "105" "533" "634")
```

```
1 (def gen-datetime
2   (gen/fmap #(java.time.Instant/ofEpochMilli %)
3             gen/large-integer))
4
5 (gen/sample gen-datetime)
6 (#inst "1970-01-01T00:00:00.000Z"
7  #inst "1969-12-31T23:59:59.999Z"
8  #inst "1969-12-31T23:59:59.998Z"
9  #inst "1970-01-01T00:00:00.000Z"
10 #inst "1969-12-31T23:59:59.998Z"
11 #inst "1970-01-01T00:00:00.007Z"
12 #inst "1970-01-01T00:00:00.000Z"
13 #inst "1969-12-31T23:59:59.938Z"
14 #inst "1969-12-31T23:59:59.999Z"
15 #inst "1970-01-01T00:00:00.035Z")
```

```
1 (def gen-metadata
2   (gen/hash-map :permissions gen-permissions-octal
3                 :user-id      gen/large-integer
4                 :group-id     gen/large-integer
5                 :created-at   gen-datetime
6                 :modified-at  gen-datetime))
7
8 (gen/generate gen-metadata)
9 {:permissions "161",
10  :user-id     484817,
11  :group-id    10350453,
12  :created-at  #inst "1970-01-01T00:01:03.269Z",
13  :modified-at #inst "1970-01-01T00:00:00.009Z"}
```

```
1 (defn gen-directory-of
2   [gen-content]
3   (gen/map gen-file-name
4             (gen/hash-map :metadata gen-metadata
5                           :content  gen-content)))
6
7 (def gen-directory
8   ;; use gen/such-that to filter out
9   ;; top-level byte arrays
10  (gen/such-that map?
11                (gen/recursive-gen
12                 gen-directory-of
13                 gen-file-contents)))
```

# gen-directory

```
1 (gen/generate gen-directory 10)
2
3 {"eqn" {:metadata {:created-at #inst "1970-01-01T00:00:00.003Z",
4           :group-id 0,
5           :modified-at #inst "1969-12-31T23:59:59.999Z",
6           :permissions "045",
7           :user-id 0},
8       :content {")" {:metadata {:created-at #inst "1969-12-31T23:59:59.999Z",
9           :group-id 0,
10          :modified-at #inst "1970-01-01T00:00:00.003Z",
11          :permissions "760",
12          :user-id 0},
13         :content #bytes "c5a1159d"}}}},
14 "xu_" {:metadata {:created-at #inst "1970-01-01T00:00:00Z",
15           :group-id 0,
16           :modified-at #inst "1969-12-31T23:59:59.999Z",
17           :permissions "372",
18           :user-id 0},
19       :content {"M" {:metadata {:created-at #inst "1969-12-31T23:59:59.999Z",
20           :group-id -1,
21           :modified-at #inst "1970-01-01T00:00:00.003Z",
```

```
1 (def a-good-directory *1)
```



```
1 (def gen-directory-with-changes  
2   ????)
```

```
1 (defn gen-changes
2   [directory]
3   ?????)
4
5 (def gen-directory-with-changes
6   ?????)
```

```
1 (defn gen-changes
2   [directory]
3   ?????)
4
5 (def gen-directory-with-changes
6   (gen/bind gen-directory
7             (fn [directory]
8               (gen/fmap (fn [changes]
9                           {:directory directory
10                            :changes changes})))
11   (gen-changes directory))))
```

```
1 (def gen-directory-with-changes
2   (gen/bind gen-directory
3     (fn [directory]
4       (gen/fmap (fn [changes]
5                   {:directory directory
6                    :changes changes}))
7       (gen-changes directory))))))
8 ;; same as
9 (def gen-directory-with-changes
10  (gen/let [directory gen-directory
11           changes (gen-changes directory)]
12    {:directory directory
13     :changes changes}))
```

```
1 (defn gen-changes
2   [directory]
3   (gen/vector (gen-appension directory)))
```

```
1 (defn gen-appension
2   [directory]
3   ;; 1. get all filepaths
4   ;; 2. select (generate) a filepath
5   ;; 3. generate random bytes to append
6   ;; 4. package together and return
7   )
```

```
1 (defn all-filepaths
2   [directory]
3   (apply concat
4     (for [[name {:keys [content]}] directory]
5       (if (map? content)
6         (map #(str name "/" %)
7              (all-filepaths content))
8         [name]))))
9
10 (all-filepaths a-good-directory)
11 ("eqn/") "xu_/M" "xu_/X")
```

```
1 (defn gen-file-appension
2   [directory]
3   (let [filepaths (all-filepaths directory)]
4     (assert (not-empty filepaths))
5     (gen/fmap (fn [[filepath more-bytes]]
6                 {:filepath filepath
7                  :append  more-bytes})
8               (gen/tuple (gen/elements filepaths)
9                           gen/bytes))))
10
11 (gen/generate
12  (gen-file-appension a-good-directory))
13 {:filepath "xu_/X",
14  :append #bytes "3b5cf eb016458d"}
```



All together now...

# gen-changes

```
1 (defn gen-changes
2   [directory]
3   ;; do something different if it's empty
4   (gen/vector (gen-file-appension directory)))
5
6 (def gen-directory-with-changes
7   (gen/let [directory gen-directory
8             changes (gen-changes directory)]
9     {:directory directory
10      :changes changes}))
11
12 (gen/generate gen-directory-with-changes 3)
13 {:directory {"" {:metadata {:permissions "217",
14                             :user-id -1,
15                             :group-id 0,
16                             :created-at #inst "1970-01-01T00:00:00.000Z",
17                             :modified-at #inst "1969-12-31T23:59:59.999Z",
18                             :content #bytes ""}},
19 :changes [{:filepath "", :append #bytes ""}]}}
```

# Summary: Building Things

Just like FP, building generators takes practice


# Fine-Tuning

# Fine-tune what, exactly?

# Fine-tune what, exactly?

- Distribution/Growth
- Shrinking

# Fine-Tuning Distribution/Growth



Distribution of gen/large-integer for size=0..9



# What does size mean exactly?

Its meaning is determined by its use:

- Test runs use  $0, 1, 2, \dots, 199, 0, 1, 2, \dots$

- Testing with `:num-tests < 200`
- `(gen/sample g)` is `size=0..9`
- `(gen/generate g)` is `size=30`

"Big" is relative.

```
1 (def g (gen/vector gen/string-ascii))
2
3 (repeatedly 5 #(gen/generate g 10))
4
5 (["V5N8'!)/" "_{^}X2m1"]
6 ["<h>YmK^$2" "" "" "B-c>M/" ""]
7 ["b.e2I*$76"]
8 ["Vo" "Sk"]
9 ["u"])
```

# string vectors

```
(gen/generate g 200)
```

```
["s9&\" (5y?%{X6AU2dkM\"m0kb+1c(-mb@)C)n.$)g|/joLvU~W,g~L"  
"&_#bCsH7A-~\"_V=(r(i|G-'9bF.)N*NYass;*swt(g06Y7XRZ37)6aQ:)lm~xgq(%xKIuZ>2V<+4dxsW7[eNyGK!Ob{y}R/o/>)(E)  
".{?T-@psGc%VDRq{R;>5u0*/OdWfN=\\);}3H}Ni>VOY!R63C@~LdVZ?WOnAs|HemOv0("'  
"$M(2ppm'H|*x$Z,0m39<2WbXbwib=-_V2o~Jv-yT~~7BRci)$?vc8/1)"'  
"u~0>dYQG7>849<%DHT11iFUk>TcUMJ'hkk*\"M\\KX1.shq8ExD=KsQ7J(mI{i5Mh8<wG'n{eL9e%c'K){,*PtYDW|pKU?,uouDsK  
"GoP>[iISny*wEGjBB^)+@(GW1$)h%aR.{JYMKq*X. ]C6ty8e#%Y9.e:>GIE(RN;f2A\\5/om?o%)"  
"h1=Z;6cJed'c\"%_G+H.qg?@oS>"  
"x[I]i1K+te6-}KTqQU<c1"  
"%5$\\u(i,>uwWz'evt~Q?m%H/t$~UB+_GpCdrTV?g3YJM2u$C,&1f.2Ym\\Q '18H>]a35FA4%Ih,\\e~Bv\\!Qfr6Kc_;>T5'9D  
"&_w5s69S%M:ZWU~q~hoYva~_$_hDiraW~auq0>P#G-].z 55w&"  
" '-OM=-uP];m8U?Q2(PsuPEp{F-NV'liq_\\,[K8.ABAtQI'MoqbU' _jGXz!$t$qm};j~V!1'Y+GU6*q7b7v-y\\MR,*"  
"i<d-b0_g(m~iMkLq4)@#srM9C(g&{'J%bz.sA+BNX~OCj\"J&}&).oMFA.%-tyBG|bCZ%h1D,]*R~{2C,"  
"pl]pVTI wVD4u)t)1,Et&})r9Tj,; 7+XzC8GrhW8Ioc41102s?+UeeW>U(1x\"c<<1hSts3ak.LbX3g}{;sI6w_','C01*s)rbA:  
"2$XMP\\\"GE;xTkLBU}RX@LQ S/S7{ziMi%GEOZYI'drF<5oDTi5&EQgN;10Ne}};c2~90<JRL<%T_ArUivoHI\\$%;"  
"LYmx,_u;, &n4%:>U-q>WCL]WGg)06}w"  
"H\"EHYMPHD6UP, _@v04kd9rDeK]Tk!H]M5=xSv3(e)_W3Y3OR%zkI'^GfGxAR.YIU00UW'-CjVZQ.Duw3$8=;tq#Tje.Y"  
"AU(%)Be>)WJZ4q)4Q:P?Z?!=Mh>[>>wfK7q&H"  
"Zpw|dR|{C\\e&Xfn*|,K@|. N.6W?N'>p5N:>UBLx4h,G5xRBPM^!Fz~zm]zMV[ufJg.\\'6wLjBr{#'5t)1zkkbHk=!C$&5B-m<g  
"A~'0h?F@LG/)+-1\\ds1+X{Inofdj;:z+=9S+/SuaLZM<rXf&u'Z>--7YJ94Zzy[:.e;uDISyDIsr4'iB(rxix#E)Ck[H*c_tEiYw*  
">r._=3&E1Z.<T(hW\"2L%UG:ur=EmWA;60wE2zPbfW19|S(*!E=8K5snV!O>T='$=mci&t&9T.WK[z9\"X'. 'VHYW! =D!XhC~VF6  
"Z{lnp|C|F.b*H+?'wgDefL/<=YHsRt<U=)vX{bk@NSR[UWJx&%J':uY4tu![] FIB8s6)9Pr@[*Y9h8st7z}\\\"pQ%7s,V] 6{/bZM)M  
"Z\"5.e#Cew{k5>~m?YyUSDTL6;.^g)Qx(J2I$E5.g@!vM|->3 4\\\"cE)?LNW604Q>5P}m&A#."  
"* B~NaUS/DhC7Z&A<:/7cm\"A>V7$WX&=\\v&-{Vn=va?[>Z(%o N%C)t a7%/_,1U3dn{a8.SP\\\"8'/t"  
"iTMn}DK\\\"l]d6Hk6nVn2U[+bJ6Tk?sG2q&*&GcGDTjJ{v//9TUTj[3\\V:'R! =)8Tguw{<}#-}Ja2f7M=4CQC>9,q<h76n+;Y43,
```

```
1 (def g-2 (gen/scale #(/ % 20)
2           (gen/vector gen/string-ascii)))
3
4 (repeatedly 5 #(gen/generate g-2 200))
5
6 (["Qw+b"
7  ".S." "C'NL.k2"
8  []
9  "Sh?7?Wv" "" "@,~d_~*z" "z<" "]Q!" "XPVS"
10 "/|@)WYvbH" "(=jTr" "ZQ}k[as;o0" "p&ri~;f1Q6"
11 ["0=]{-=\\" "BTqjqWg" ":Nlz" "F6I*(P)"])
```



test.check / TCHECK-106

## Variable-sized collection generators have exponential sizing issues when composed

# Too Weird / Not Weird Enough



# Too Weird / Not Weird Enough

```
1  ;; doesn't often generate
2  ;; nontrivial text files
3  (def gen-file-contents
4    gen/bytes)
```

What's the best amount of weird?

# Too Weird / Not Weird Enough

```
1  ;; doesn't often generate
2  ;; nontrivial text files
3  (def gen-file-contents
4    gen/bytes)
5
6  ;; half random bytes, half UTF-8
7  (def gen-file-contents-2
8    (gen/one-of [gen/bytes
9                 gen-UTF8-bytes]))
```

# Too Weird / Not Weird Enough

```
1  ;; Only generates small files
2  (def gen-file-contents-2
3    (gen/one-of [gen/bytes
4                 gen-UTF8-bytes]))
```

# Too Weird / Not Weird Enough

```
1  ;; Only generates small files
2  (def gen-file-contents-2
3    (gen/one-of [gen/bytes
4                 gen-UTF8-bytes]))
5
6  ;; Occasionally generates big files!
7  (def gen-file-contents-3
8    (gen/frequency [[45 gen/bytes]
9                   [45 gen-UTF8-bytes]
10                  [10 (gen/scale
11                       (fn [size]
12                         (if (<= 100 size)
13                             (* size size)
14                             size))
15                       gen/bytes)]]))
```

```
1 (def gen-datetime
2   (gen/fmap #(java.time.Instant/ofEpochMilli %)
3             gen/large-integer))
4
5 (gen/sample gen-datetime)
6 (#inst "1970-01-01T00:00:00.000Z"
7  #inst "1969-12-31T23:59:59.999Z"
8  #inst "1969-12-31T23:59:59.998Z"
9  #inst "1970-01-01T00:00:00.000Z"
10 #inst "1969-12-31T23:59:59.998Z"
11 #inst "1970-01-01T00:00:00.007Z"
12 #inst "1970-01-01T00:00:00.000Z"
13 #inst "1969-12-31T23:59:59.938Z"
14 #inst "1969-12-31T23:59:59.999Z"
15 #inst "1970-01-01T00:00:00.035Z")
```

```
1 (def gen-datetime-components
2   (gen/hash-map
3     :year    (gen/fmap #(+ % 2017) gen/int)
4     :month   (gen/large-integer* {:min 1, :max 12})
5     :day     (gen/large-integer* {:min 1, :max 31})
6     :hour    (gen/large-integer* {:min 0, :max 23})
7     :minute  (gen/large-integer* {:min 0, :max 59})
8     :second  (gen/large-integer* {:min 0, :max 59})
9     :millis  (gen/large-integer* {:min 0, :max 1000})))
```

```

1  (defn construct-datetime
2    [{:keys [year month day
3            hour minute second millis]}]
4    (try
5      (java.time.Instant/parse
6        (format "%04d-%02d-%02dT%02d:%02d:%02d.%03dZ"
7              year month day
8              hour minute second millis))
9      (catch Exception e
10         ;; kind of dumb, but it works and it's easy
11         (java.time.Instant/parse
12           (format "%04d-%02d-%02dT%02d:%02d:%02d.%03dZ"
13                 year month 28
14                 hour minute second millis))))))

```



```
1 (def gen-datetime-2
2   (gen/fmap construct-datetime
3             gen-datetime-components))
4
5 (gen/sample gen-datetime-2)
6 (#inst "2017-02-02T01:01:01.001Z"
7  #inst "2016-01-02T01:01:01.000Z"
8  #inst "2017-02-01T00:01:00.001Z"
9  #inst "2015-02-01T00:03:01.003Z"
10 #inst "2016-07-02T01:00:00.002Z"
11 #inst "2022-03-02T01:02:03.001Z"
12 #inst "2020-05-03T04:00:01.000Z"
13 #inst "2019-02-09T10:05:11.000Z"
14 #inst "2020-04-08T12:27:04.053Z"
15 #inst "2013-08-06T21:04:03.001Z")
```

# Fine-Tuning Shrinking

gen-datetime vs gen-datetime-2

```
1 (def ^{:added "0.9.0"} uuid
2   "Generates a random type-4 UUID. Does not shrink."
3   (no-shrink
4     ...))
```

```
1 (def gen-matrix
2   (gen/let [width gen/nat
3             rows (gen/vector
4                   (gen/vector gen/large-integer
5                               width))]
6     rows))
7
8 (quick-check 10000
9   (prop/for-all [matrix gen-matrix]
10    (->> matrix
11      (apply concat)
12      (not-any? #{42}))))
13
14 ;; fails on a large matrix, shrinks to [[0 42 0]]
```

# Custom Shrinking

## Summary: Tweaking Things

- Things can be too big or small; `gen/scale` can help
- You can target specific sensitivities with `gen/one-of` and `gen/frequency`
- Modeling the domain better can help
- Shrinking is hard

Welp!



# That's About It

- Generator combinators are abstract and declarative, to support growth and shrinking
- With practice, you can generate anything, and customize its distribution, growth, and shrinking

# Thanks!